

ЗАДАНИЕ №18

6 минут

ТИП 1

работа с таблицей

ФУНКЦИИ, КОТОРЫЕ НУЖНО ЗНАТЬ

ФУНКЦИЯ	НАЗВАНИЕ (РУС. ЯЗ.)	НАЗВАНИЕ (АНГЛ. ЯЗ.)
остаток от деления	ОСТ(делимое; делитель)	MOD(делимое; делитель)
условие	ЕСЛИ(условие; если_истина; если_ложь)	IF(условие; если_истина; если_ложь)
целочисленное деление	ЧАСТНОЕ(делимое; делитель)	QUOTIENT(делимое; делитель)
максимум	МАКС(ряд чисел)	MAX(ряд чисел)
минимум	МИН(ряд чисел)	MIN(ряд чисел)
количество	СЧЁТ(ряд чисел)	COUNT(ряд чисел)
количество, удовлетворяющих одному условию	СЧЁТЕСЛИ(ряд чисел; условие)	COUNTIF(ряд чисел; условие)
количество, удовлетворяющих нескольким условиям	СЧЁТЕСЛИМН(ряд чисел 1; условие 1; ряд чисел 2; условие 2...)	COUNTIFS(ряд чисел 1; условие 1; ряд чисел 2; условие 2...)
максимальное, удовлетворяющее нескольким условиям	МАКСЕСЛИ(ряд чисел; ряд 1, на который действует условие; условие 1)	MAXIFS(ряд чисел; ряд 1, на который действует условие; условие 1)
минимальное, удовлетворяющее нескольким условиям	МИНЕСЛИ(ряд чисел; ряд 1, на который действует условие; условие 1 ...)	MINIFS(ряд чисел; ряд 1, на который действует условие; условие 1 ...)

1.1: РАЗБОР (ДЕМОВЕРСИЯ)

Исходные данные записаны в файле **1_1.xls** в виде электронной таблицы прямоугольной формы. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя из левой верхней клетки в правую нижнюю. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

1.1: РАЗБОР (ДЕМОВЕРСИЯ)

План решения:

Для начала определим максимально возможную набранную сумму:

- 1) открыть файл 1_1.xls; скопировать табличку с данными, вставить ниже, залить вставленную табличку каким-то цветным фоном и удалить из неё уданные;
- 2) заполнить самую верхнюю строчку во вставленной табличке;
- 3) заполнить самый левый столбец во вставленной табличке;
- 4) заполнить все остальные ячейки во вставленной табличке.

Для определения минимально возможно набранной суммы потребуется сделать всё то же самое.

1.1: РАЗБОР (ДЕМОВЕРСИЯ)

Заливаем вставленную табличку
цветным фоном:


	A	B	C	D	E	F	G	H	I	J	K
1	15	43	23	44	85	19	72	67	36	80	
2	100	48	92	64	91	61	71	2	82	74	
3	69	9	12	9	91	52	94	58	69	42	
4	53	42	18	58	93	55	48	76	18	13	
5	49	63	16	13	45	6	81	19	53	63	
6	49	53	23	44	61	2	2	63	88	16	
7	2	75	100	27	72	64	95	49	51	84	
8	64	68	17	16	18	2	30	57	86	6	
9	64	84	53	19	23	91	82	2	2	59	
10	72	47	56	85	9	26	47	48	91	91	
11											
12											
13	15	43	23	44	85	19	72	67	36	80	
14	100	48	92	64	91	61	71	2	82	74	
15	69	9	12	9	91	52	94	58	69	42	
16	53	42	18	58	93	55	48	76	18	13	
17	49	63	16	13	45	6	81	19	53	63	
18	49	53	23	44	61	2	2	63	88	16	
19	2	75	100	27	72	64	95	49	51	84	
20	64	68	17	16	18	2	30	57	86	6	
21	64	84	53	19	23	91	82	2	2	59	
22	72	47	56	85	9	26	47	48	91	91	
23											
24											
25											

The screenshot shows the Microsoft Excel interface. The 'Home' tab is selected and circled in red. The 'Theme Colors' palette is open, also circled in red, showing a grid of color swatches. The spreadsheet below shows the same data as the previous image, but with the cells from row 13 to row 22 highlighted in yellow.

1.1: РАЗБОР (ДЕМОВЕРСИЯ)

Заполняем ЛЕВУЮ ВЕРХНЮЮ ячейку (т.к. по условию движение начинается из левого верхнего угла и заканчивается в правом нижнем).

Для этого просто копируем значение из A1:



	A	B	C	D	E	F
1	15	43	23	44	85	19
2	100	48	92	64	91	61
3	69	9	12	9	91	52
4	53	42	18	58	93	55
5	49	63	16	13	45	6
6	49	53	23	44	61	2
7	2	75	100	27	72	64
8	64	68	17	16	18	2
9	64	84	53	19	23	91
10	72	47	56	85	9	26
11						
12						
13	15					
14						
15						
16						

1.1: РАЗБОР (ДЕМОВЕРСИЯ)

Робот может двигаться вправо и вниз. Из левой верхней ячейки двигаясь только вправо, Робот проходит всю верхнюю строчку. Заполним её суммами, которые наберёт Робот, если дойдёт до этой ячейки:


	A	B	C	D	E
1	15	43	23	44	85
2	100	48	92	64	91
3	69	9	12	9	91
4	53	42	18	58	93
5	49	63	16	13	45
6	49	53	23	44	61
7	2	75	100	27	72
8	64	68	17	16	18
9	64	84	53	19	23
10	72	47	56	85	9
11					
12					
13	15	=A13+B1			
14					

	A	B	C	D	E
1	15	43	23	44	85
2	100	48	92	64	91
3	69	9	12	9	91
4	53	42	18	58	93
5	49	63	16	13	45
6	49	53	23	44	61
7	2	75	100	27	72
8	64	68	17	16	18
9	64	84	53	19	23
10	72	47	56	85	9
11					
12					
13	15	58	=B13+C1		
14					

1.1: РАЗБОР (ДЕМОВЕРСИЯ)

Если Робот будет двигаться только вниз,
он пройдёт весь левый столбец.

Заполним левый столбец суммами,
которые наберёт Робот, если дойдёт
до соответствующей ячейки.



	A	B	C	D	E
1	15	43	23	44	
2	100	48	92	64	
3	69	9	12	9	
4	53	42	18	58	
5	49	63	16	13	
6	49	53	23	44	
7	2	75	100	27	
8	64	68	17	16	
9	64	84	53	19	
10	72	47	56	85	
11					
12					
13	15	58	81	125	2
14	=A13+A2				
15					
16					

1.1: РАЗБОР (ДЕМОНСТРАЦИЯ)

Заполним оставшиеся ячейки. Сперва будем искать наибольшую возможную сумму. Формула для ячейки во втором столбце второй строки:


	A	B	C	D	E	F	G
1	15	43	23	44	85	19	72
2	100	48	92	64	91	61	71
3	69	9	12	9	91	52	94
4	53	42	18	58	93	55	48
5	49	63	16	13	45	6	81
6	49	53	23	44	61	2	2
7	2	75	100	27	72	64	95
8	64	68	17	16	18	2	30
9	64	84	53	19	23	91	82
10	72	47	56	85	9	26	47
11							
12							
13	15	58	81	125	210	229	301
14	115	=MAX(A14;B13)+B2					
15	184						
16	237						

1.1: РАЗБОР (ДЕМОВЕРСИЯ)

Т.е. максимально возможная сумма – 1320.

Для поиска минимально возможной суммы менять левый столбец и верхнюю строчку **не нужно**.

Потребуется заменить только формулу с максимумом: взять вместо него минимум.



	A	B	C	D	E	F
1	15	43	23	44	85	19
2	100	48	92	64	91	61
3	69	9	12	9	91	52
4	53	42	18	58	93	55
5	49	63	16	13	45	6
6	49	53	23	44	61	2
7	2	75	100	27	72	64
8	64	68	17	16	18	2
9	64	84	53	19	23	91
10	72	47	56	85	9	26
11						
12						
13	15	58	81	125	210	229
14	115	=MIN(A14;B13)+B2				
15	184					
16	237					

1.2: САМОСТОЯТЕЛЬНО

Исходные данные записаны в файле **1_2.xls** в виде электронной таблицы прямоугольной формы. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя из левой верхней клетки в правую нижнюю. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

1.2: САМОСТОЯТЕЛЬНО

Ответ: 1357 701

1.3: САМОСТОЯТЕЛЬНО

Исходные данные записаны в файле **1_3.xls** в виде электронной таблицы прямоугольной формы. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя из **левой НИЖНЕЙ** клетки в **правую ВЕРХНЮЮ**. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

1.3: САМОСТОЯТЕЛЬНО

Ответ: 1236 717

1.4: РАЗБОР

Исходные данные для Робота записаны в файле **1_4.xls** в виде электронной таблицы прямоугольной формы. Робот может двигаться только вниз и вправо. Робот может брать монеты **только с тех клеток, где количество монет нечётно**. Если количество монет чётно, то Робот не берёт в этой клетке ни одной монеты. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя из левой верхней клетки в правую нижнюю. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.


1.4: РАЗБОР

Задача отличается от предыдущих только тем, что все чётные числа в таблице нужно заменить на нули. Для этого скопируем таблицу, вставим её ниже, изменим фон и удалим все данные. После этого запишем в левую верхнюю

ячейку формулу:

`=IF(MOD(A1;2) = 1;A1;0)`

Если в ячейке находится нечётное число (остаток от деления на 2 равен 1), то сохраняем значение в ячейке, иначе – заменяем на 0.



The screenshot shows an Excel spreadsheet with a table of numbers. The formula bar at the top displays `=IF(MOD(A1;2) = 1;A1;0)`. The table has 15 rows and 10 columns (A-J). The first 13 rows contain original data. Row 14 is highlighted in yellow and contains the formula in cell A14. Row 15 is also highlighted in yellow and shows the result of the formula applied to the first row of data, where even numbers are replaced by 0 and odd numbers remain the same.

	A	B	C	D	E	F	G	H	I	J
1	43	99	79	21	70	56	54	24	42	38
2	90	38	7	99	94	34	24	40	52	4
3	9	82	98	20	88	92	76	42	41	98
4	78	32	9	27	57	20	86	86	7	58
5	84	37	70	31	26	85	25	100	79	83
6	76	32	2	48	52	74	28	33	27	27
7	13	75	34	73	25	78	33	58	71	3
8	42	93	66	22	54	24	79	98	56	46
9	97	75	26	12	92	2	91	44	66	40
10	42	86	31	86	6	22	96	62	41	42
11										
12										
13										
14	<code>=IF(MOD(A1;2) = 1;A1;0)</code>			21	0	0	0	0	0	0
15	0	0	7	99	0	0	0	0	0	0

1.4: РАЗБОР

Находить максимальную и минимальную сумму нужно для новой таблицы (делаем всё то же самое, что и в предыдущих задачах, только в формулы подставляем ячейки из новой таблицы, с нулями).

14	43	99	79	21	0	0	0	0	0	0	43	142	221	242	242	242	242	242	242	242
15	0	0	7	99	0	0	0	0	0	0	43	142	228	341	341	341	341	341	341	341
16	9	0	0	0	0	0	0	0	41	0	52	142	228	341	341	341	341	341	382	382
17	0	0	9	27	57	0	0	0	7	0	52	142	237	368	425	425	425	425	432	432
18	0	37	0	31	0	85	25	0	79	83	52	179	237	399	425	510	535	535	614	697
19	0	0	0	0	0	0	0	33	27	27	52	179	237	399	425	510	535	568	641	724
20	13	75	0	73	25	0	33	0	71	3	65	254	254	472	497	510	568	568	712	727
21	0	93	0	0	0	0	79	0	0	0	65	347	347	472	497	510	647	647	712	727
22	97	75	0	0	0	0	91	0	0	0	162	422	422	472	497	510	738	738	738	738
23	0	0	31	0	0	0	0	0	41	0	162	422	453	472	497	510	738	738	779	779
24																				
25																				
26											43	142	221	242	242	242	242	242	242	242
27											43	43	50	149	149	149	149	149	149	149
28											52	43	43	43	43	43	43	43	84	84
29											52	43	52	70	100	43	43	43	50	50
30											52	80	52	83	83	128	68	43	122	133
31											52	52	52	52	52	52	52	76	103	130
32											65	127	52	125	77	52	85	76	147	133
33											65	158	52	52	52	52	131	76	76	76
34											162	233	52	52	52	52	143	76	76	76
35											162	162	83	52	52	52	52	52	93	76
36																				

Ответ: 779

1.5: САМОСТОЯТЕЛЬНО

Исходные данные для Робота записаны в файле **1_5.xls** в виде электронной таблицы прямоугольной формы. Робот может двигаться только вниз и вправо. Робот может брать монеты **только с тех клеток, где количество монет чётно**. Если количество монет нечётно, то Робот не берёт в этой клетке ни одной монеты. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя из левой верхней клетки в правую нижнюю. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

1.5: САМОСТОЯТЕЛЬНО

Ответ: 796 52

1.6: САМОСТОЯТЕЛЬНО

Квадрат разлинован на $N \times N$ клеток ($1 < N < 17$). Исполнитель Робот может перемещаться по клеткам, выполняя за одно перемещение одну из двух команд: вправо или вниз. По команде вправо Робот перемещается в соседнюю правую клетку, по команде вниз – в соседнюю нижнюю. При попытке выхода за границу квадрата Робот разрушается. Перед каждым запуском Робота в каждой клетке квадрата лежит монета достоинством от 1 до 100. Посетив клетку, Робот **забирает монету с собой только в том случае, если её номинал – число, кратное 3; если номинал монеты – число, не кратное 3, то Робот не берёт монету**; это также относится к начальной и конечной клетке маршрута Робота. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя из левой верхней клетки в правую нижнюю. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

Исходные данные для Робота записаны в файле **1_6.xls** в виде электронной таблицы прямоугольной формы.

1.6: САМОСТОЯТЕЛЬНО

Ответ: 684 105

1.7: САМОСТОЯТЕЛЬНО

Исходные данные для Робота записаны в файле **1_7.xls** в виде электронной таблицы прямоугольной формы. Робот может двигаться только вниз и вправо. Для сбора денег у Робота есть контейнеры вместимостью **8 монет** каждый. С каждой клетки Робот забирает наибольшее количество контейнеров, **полностью заполненных** монетами. Если контейнер не заполнен до конца, а монеты в клетке кончились, робот высыпает из него монеты перед переходом в следующую клетку. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя из левой верхней клетки в правую нижнюю. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

1.7: САМОСТОЯТЕЛЬНО

Ответ: 1144 448

Подсказка: требуется создать дополнительную таблицу, в каждой ячейке которой содержится **результат целочисленного деления** исходного значения **на 8**.

1.8: РАЗБОР

Исходные данные для Робота записаны в файле **1_8.xls** в виде электронной таблицы прямоугольной формы. Роботу нужно перейти через поле с севера (верхняя строка) на юг (нижняя строка). Он может начать переход с любой клетки верхней строки и закончить на любой клетке нижней строки. С каждым шагом Робот переходит в следующий ряд и может за одно перемещение попасть в одну из трех клеток следующей строки (на клетку прямо или боковые с ней). Ходы только в бок (без смены строки) и/или назад запрещены. В каждой клетке поля лежит монета достоинством от 1 до 100. Робот собирает все монеты по пройденному маршруту.

Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя с северной границы поля (сверху) до южной границы поля (снизу). В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

1.9: САМОСТОЯТЕЛЬНО

Исходные данные для Робота записаны в файле **1_9.xls** в виде электронной таблицы прямоугольной формы. Роботу нужно перейти через поле с запада (левый столбец) на восток (правый столбец). Он может начать переход с любой клетки левого столбца и закончить на любой клетке правого столбца. С каждым шагом Робот переходит в следующий столбец и может за одно перемещение попасть в одну из трех клеток следующего столбца (на клетку прямо или боковые с ней). Ходы только в бок (без смены столбца) и/или назад запрещены. В каждой клетке поля лежит монета достоинством от 1 до 100. Робот собирает все монеты по пройденному маршруту. Определите максимальную и минимальную денежную сумму, которую может собрать Робот, пройдя с западной границы поля (слева) до восточной границы поля (справа). В ответе укажите два числа: сначала максимальную сумму, затем минимальную.

1.9: САМОСТОЯТЕЛЬНО

Ответ: 785 176

1.10: САМОСТОЯТЕЛЬНО

Квадрат разлинован на $N \times N$ клеток ($1 < N < 17$). Исполнитель Робот может перемещаться по клеткам, выполняя за одно перемещение одну из двух команд: вправо или вниз. По команде вправо Робот перемещается в соседнюю правую клетку, по команде вниз – в соседнюю нижнюю. При попытке выхода за границу квадрата Робот разрушается. Перед каждым запуском Робота в каждой клетке квадрата записано число от 10 до 99. Посетив клетку с нечетным значением, Робот увеличивает счет на 1; иначе увеличивает счёт на 2.

Определите максимальное и минимальное значение счета, который может собрать Робот, пройдя из левой верхней клетки в правую нижнюю. В ответе укажите два числа – сначала максимальную сумму, затем минимальную.

Исходные данные представляют собой электронную таблицу в файле **1_10.xls** размером $N \times N$, каждая ячейка которой соответствует клетке квадрата.

Пример входных данных:

1	8	8	4
10	1	1	3
1	3	12	2
2	3	5	6

Для данных входных данных ответом должна быть пара чисел **9 12**.

1.10: САМОСТОЯТЕЛЬНО

Ответ: 32 23

Подсказка: требуется создать дополнительную таблицу, в которой все числа будут заменены на 1 (если в исходной ячейке стояло нечётное число) и на 2 (если в исходной ячейке стояло чётное число).

ТИП 2

работа с последовательностью чисел

2.1: РАЗБОР (СТАТГРАД ОТ 22.10.20)

Дана последовательность вещественных чисел. Из неё необходимо выбрать несколько подряд идущих чисел так, чтобы каждое следующее число было меньше предыдущего. Какую максимальную сумму могут иметь выбранные числа? В ответе запишите целую часть получившейся суммы.

Например, для входных данных

3,3

5,2

5,9

1,3

1,7

4,5

максимально возможная сумма равна 7,2, в ответе надо записать число 7.

Исходные данные записаны в виде столбца электронной таблицы в файле **2_1.xls**.


2.1: РАЗБОР (СТАТГРАД ОТ 22.10.20)

Если **текущее** число меньше **предыдущего** числа, то текущее число нужно добавить в сумму накопленных чисел. Если текущее число **больше или равно** предыдущего числа, то сумма будет равна текущему числу (начинаем суммирование заново). Для первого числа сумма равна самому этому числу.

	A	B	C	D
1	75,96			
2	63,26			
3	110,70			
4	123,18			
5	138,49			
6	140,28			
7	125,36			
8	12,31			
9	128,93			
10	13,70			
11	138,19			
12	74,35			
13	12,47			

2.1: РАЗБОР (СТАТГРАД ОТ 22.10.20)


Будем записывать суммы во второй столбец (в столбец В).
Первая сумма (ячейка В1) равна самому числу в А1.



	A	B	C
1	75,96	=A1	
2	63,26		
3	110,70		
4	122,10		

2.1: РАЗБОР (СТАТГРАД ОТ 22.10.20)


Для остальных чисел действует правило: если число меньше предыдущего, добавляем его в сумму, иначе начинаем сумму заново (сбрасываем сумму до текущего числа).



	A	B	C	D	E	F	G
1	75,96	75,96					
2	63,26	=IF(A2<A1;B1;0)+A2					
3	110,70						
4	123,18						
5	138,49						


2.1: РАЗБОР (СТАТГРАД ОТ 22.10.20)

Осталось только растиражировать формулу на весь столбец В и найти максимальное число (т.е. накопленную сумму) в столбце В.



The screenshot shows an Excel spreadsheet with columns A through F and rows 1 through 7. Column A contains values: 75,96, 63,26, 110,70, 123,18, 138,49, 140,28, 125,36. Column B contains values: 75,96, 139,22, 110,70, 123,18, 138,49, 140,28, 265,64. Column D contains the text 'MAX' in row 1 and the formula '=MAX(B1:B500)' in row 2. The formula bar at the top shows '=MAX(B1:B500)'.

	A	B	C	D	E	F
1	75,96	75,96		MAX		
2	63,26	139,22		=MAX(B1:B500)		
3	110,70	110,70				
4	123,18	123,18				
5	138,49	138,49				
6	140,28	140,28				
7	125,36	265,64				



The screenshot shows the same Excel spreadsheet as the previous one, but with the result of the MAX formula. The formula bar at the top shows 'K15'. Cell D2 now contains the value '399,96'.

	A	B	C	D	E	F
1	75,96	75,96		MAX		
2	63,26	139,22		399,96		
3	110,70	110,70				
4	123,18	123,18				

Т.к. в ответ просили записать целую часть суммы, ответ – 399.

2.2: САМОСТОЯТЕЛЬНО

Дана последовательность вещественных чисел. Из неё необходимо выбрать несколько подряд идущих чисел так, чтобы каждое следующее число было меньше предыдущего. Какую максимальную сумму могут иметь выбранные числа? В ответе запишите целую часть получившегося числа.

Исходные данные записаны в виде столбца электронной таблицы в файле **2_2.xls**.

2.2: САМОСТОЯТЕЛЬНО

Ответ: 608

2.3: САМОСТОЯТЕЛЬНО

Дана последовательность вещественных чисел. Из неё необходимо выбрать несколько подряд идущих чисел так, чтобы каждое следующее число было **больше предыдущего**. Какую максимальную сумму могут иметь выбранные числа?

Исходные данные записаны в виде столбца электронной таблицы в файле **2_3.xls**.

2.3: САМОСТОЯТЕЛЬНО

Ответ: 617

2.4: САМОСТОЯТЕЛЬНО

Дана последовательность вещественных чисел. Из неё необходимо выбрать несколько подряд идущих чисел так, чтобы каждое следующее число отличалось от предыдущего **не более чем на 2**. Какую максимальную сумму могут иметь выбранные числа? В ответе запишите только целую часть максимально возможной суммы. Исходная последовательность записана в виде одного столбца электронной таблицы в файле **2_4.xls**.

2.4: САМОСТОЯТЕЛЬНО

Ответ: 58

Подсказка: в формуле для тиражирования условие

$$A2 < A1$$

нужно заменить на

$$\text{ABS}(A2 - A1) \leq 2$$

2.5: САМОСТОЯТЕЛЬНО

Дана последовательность вещественных чисел. Из неё необходимо выбрать несколько подряд идущих чисел так, чтобы каждое следующее число отличалось от предыдущего **не менее** чем на 16. Какую максимальную сумму могут иметь выбранные числа? В ответе запишите только целую часть максимально возможной суммы. Исходная последовательность записана в виде одного столбца электронной таблицы в файле **2_5.xls**.

2.5: САМОСТОЯТЕЛЬНО

Ответ: 33

2.6: РАЗБОР

Дана последовательность натуральных чисел.



Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не менее чем на 7.**

Определите количество таких пар, для которых сумма чисел находится в диапазоне от 1500 до 2000, включая 1500 и 2000.

Исходные данные записаны в виде столбца электронной таблицы в файле **2_6.xls**.

2.6: РАЗБОР

Для каждого числа определим, какие минимальное и максимальное числа ему можно поставить в пару. Минимальное число будет равно (1500 – текущее число), а максимальное будет равно (2000 – текущее число).



Формулы, естественно, нужно будет растиражировать на весь столбец B и весь столбец C.


2.6: РАЗБОР

Теперь посчитаем количество пар. Если X является парой Y , то и Y является парой X , т.е. если мы для каждого числа найдём каждую пару, то мы посчитаем пары **дважды**. Поэтому мы будем находить пары только среди **предыдущих** чисел. Так каждая пара будет посчитана только один раз.

Для первых 7 чисел предыдущих пар нет

2.6: РАЗБОР

Для 8-го числа запишем формулу



	A	B	C	D	E	F	G
1	123	1377	1877	0			
2	985	515	1015	0			
3	73	1427	1927	0			
4	254	1246	1746	0			
5	438	1062	1562	0			
6	856	644	1144	0			
7	424	1076	1576	0			
8	925	575	1075	=COUNTIFS(\$A\$1:A1;">="&B8;\$A\$1:A1;"<="&C8)			


Условий два (число не меньше минимального и не больше максимального), поэтому используем функцию COUNTIFS, которая позволяет задавать несколько условий.

Все числа перебираем, начиная с первого, поэтому верхняя граница – абсолютная ссылка, \$A\$1. Нижняя граница – ячейка, которая находится на 7 ячеек выше данной. Нижняя граница смещается, поэтому ссылка будет относительной.

Для ячейки D8 верхняя и нижняя границы совпадают. В остальных ячейках при тиражировании формулы нижняя граница будет автоматически смещаться.

2.6: РАЗБОР

Пример формулы для ячейки D15:



	A	B	C	D	E	F	G
1	123	1377	1877	0			
2	985	515	1015	0			
3	73	1427	1927	0			
4	254	1246	1746	0			
5	438	1062	1562	0			
6	856	644	1144	0			
7	424	1076	1576	0			
8	925	575	1075	0			
9	182	1318	1818	0			
10	421	1079	1579	0			
11	101	1399	1899	0			
12	693	807	1307	1			
13	233	1267	1767	0			
14	137	1363	1863	0			
15	810	690	1190	=COUNTIFS(\$A\$1:A8;">="&B15;\$A\$1:A8;"<="&C15)			
16	904	500	1000	0			

Менять формулу вручную для каждой ячейки не нужно, она меняется сама при тиражировании!

2.6: РАЗБОР

Осталось только посчитать количество пар. Это сумма всех чисел в столбце D

The image shows two screenshots of an Excel spreadsheet. The top screenshot shows the formula bar with the formula `=SUM(D1:D1000)` and the spreadsheet grid with columns A through G and rows 1 through 3. The bottom screenshot shows the same spreadsheet with the formula bar still displaying `=SUM(D1:D1000)`, but now the cell F1 contains the result 58963.

	A	B	C	D	E	F	G
1	123	1377	1877	0		=SUM(D1:D1000)	
2	985	515	1015	0			
3	73	1427	1927	0			

	A	B	C	D	E	F	G
1	123	1377	1877	0		58963	
2	985	515	1015	0			
3	73	1427	1927	0			

Ответ: 58963

2.7: САМОСТОЯТЕЛЬНО

Дана последовательность натуральных чисел.

Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не менее чем на 3.**

Определите максимальную сумму чисел среди таких пар. Исходные данные записаны в виде столбца электронной таблицы в файле **2_7.xls**.

2.7: САМОСТОЯТЕЛЬНО

Ответ: 2000

Подсказка: ищем максимальное число среди предыдущих чисел (которые выше текущего минимум на 3 ячейки). Первые три ячейки в столбце В заполняем нулями, далее тиражируем формулу из B4.

П

	A	B	C	D	E
1	123	0		2000	
2	985	0			
3	73	0			
4	254	=MAX(\$A\$1:A1)+A4			
5	120	1173			

Максимум среди

	A	B	C	D	E
1	123	0		=MAX(B1:B1000)	
2	985	0			
3	73	0			

2.8: РАЗБОР


Дана последовательность натуральных чисел.

Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не более чем на 5**.

Определите количество таких пар, для которых сумма чисел находится в диапазоне от 1000 до 1500, **не включая** 1000 и 1500. Исходные данные записаны в виде столбца электронной таблицы в файле **2_8.xls**.


2.8: РАЗБОР

Для решения задачи также потребуется создать два столбца, в которых для каждого числа из столбца А будет записано минимальное подходящее число и максимальное подходящее число.



The screenshot shows the Excel formula bar with the formula $=1000-A1+1$. Below the formula bar, a spreadsheet grid is visible with columns A, B, and C. Row 1 contains the value 123 in column A, and the formula $=1000-A1+1$ is entered in column B. Row 2 contains the value 985 in column A.

	A	B	C
1	123	$=1000-A1+1$	
2	985		




The screenshot shows the Excel formula bar with the formula $=1500-A1-1$. Below the formula bar, a spreadsheet grid is visible with columns A, B, and C. Row 1 contains the value 123 in column A, the value 878 in column B, and the formula $=1500-A1-1$ is entered in column C. Row 2 contains the value 985 in column A.

	A	B	C
1	123	878	$=1500-A1-1$
2	985		

-1 и +1 дописываем в формулы, т.к. границы **не включены**.
Тиражируем формулы в столбцах В и С.

2.8: РАЗБОР

В столбце D будем определять количество подходящих пар.
Для первого числа количество пар равно 0.



	A	B	C	D
1	123	878	1376	0
2	985	16	514	
3	73	928	1426	

2.8: РАЗБОР


Для следующих 4-х чисел верхняя граница в формуле **зафиксирована** (используется абсолютная ссылка на ячейку A1, т.е. \$A\$1), нижняя граница не фиксирована, т.к. сверху чисел не хватает: для строк с номерами 2-5 нет строки выше, у которой порядковый номер отличается более чем на 4.

	A	B	C	D	E	F	G
1	123	878	1376	0			
2	985	16	514	=COUNTIFS(\$A\$1:A1;">="&B2;\$A\$1:A1;"<="&C2)			
3	73	928	1426				
4	254	747	1245				
5	438	563	1061				

	A	B	C	D
1	123	878	1376	0
2	985	16	514	1
3	73	928	1426	1
4	254	747	1245	1
5	438	563	1061	1
6	856	145	643	1
7	424	577	1075	1

2.8: РАЗБОР

Для всех остальных строк используется формула, в которой обе границы **не зафиксированы** (т.е. ссылки на ячейки не абсолютные, а относительные, без знаков \$). Тогда диапазон ячеек, среди которых ищем пары, будет меняться, и мы будем рассматривать только 5 предыдущих чисел, а не все числа с самого начала.




	A	B	C	D	E	F	G
1	123	878	1376	0			
2	985	16	514	1			
3	73	928	1426	1			
4	254	747	1245	1			
5	438	563	1061	1			
6	856	145	643	=COUNTIFS(A1:A5;">="&B6;A1:A5;"<="&C6)			
7	424	577	1075				
8	925	76	574				
9	182	819	1317				


Полученную формулу тиражируем на все оставшиеся ячейки в столбце D.

2.8: РАЗБОР

Осталось только посчитать общее количество пар – просуммировать все числа в столбце D.



	A	B	C	D	E	F
1	123	878	1376	0	=SUM(D1:D1000)	
2	985	16	514	1		
3	73	928	1426	1		
4	254	747	1245	1		
5	438	563	1061	1		
6	856	145	643	2		
7	424	577	1075	2		



	A	B	C	D	E
1	123	878	1376	0	1874
2	985	16	514	1	
3	73	928	1426	1	
4	254	747	1245	1	

Ответ: 1874

2.9: САМОСТОЯТЕЛЬНО

Дана последовательность натуральных чисел.

Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не более чем на 5.**

Определите количество таких пар, для которых сумма чисел меньше 100. Исходные данные записаны в виде столбца электронной таблицы в файле **2_9.xls**.

2.9: САМОСТОЯТЕЛЬНО

Ответ: 15

2.10: РАЗБОР


Дана последовательность натуральных чисел.

Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не менее чем на 9.**

Определите количество таких пар, для которых сумма чисел нечётная. Исходные данные записаны в виде столбца электронной таблицы в файле **2_10.xls**.

2.10: РАЗБОР

В этой задаче нужно учитывать остаток от деления числа на 2, поэтому в столбце В для каждого числа запишем, чему равен остаток от деления числа на 2:




Формулу растиражируем на весь столбец В.

2.10: РАЗБОР

Теперь в столбце С посчитаем, сколько чисел из предыдущих имеют НЕ такой остаток от деления как текущее число. Т.е. в столбце С должны стоять **разные** числа. Условие одно, поэтому можно будет использовать функцию COUNTIF.


Первые 9 строк в столбце С заполняем нулями, т.к. по условию им не подходит ни одно предыдущее число



	A	B	C
1	123	1	0
2	985	1	0
3	73	1	0
4	254	0	0
5	438	0	0
6	856	0	0
7	424	0	0
8	925	1	0
9	182	0	0

2.10: РАЗБОР

В ячейке C10 пишем формулу для подсчёта пар:



The screenshot shows an Excel spreadsheet with columns A through E and rows 1 through 10. The formula bar at the top displays the formula `=COUNTIF(B1:B1;"<>"&B10)`. The spreadsheet data is as follows:

	A	B	C	D	E
1	123	1	0		
2	985	1	0		
3	73	1	0		
4	254	0	0		
5	438	0	0		
6	856	0	0		
7	424	0	0		
8	925	1	0		
9	182	0	0		
10	421	1	<code>=COUNTIF(\$B\$1:B1;"<>"&B10)</code>		

Формулу нужно будет растиражировать на весь столбец C.

2.10: РАЗБОР

Теперь в столбце С написано количество пар для каждого числа. Осталось их только сложить:

	A	B	C	D	E	F	G
1	123	1	0			=SUM(C1:C1000)	
2	985	1	0				
3	73	1	0				
4	254	0	0				
5	438	0	0				
6	856	0	0				
7	424	0	0				
8	925	1	0				

	A	B	C	D	E	F	G
1	123	1	0			245937	
2	985	1	0				
3	73	1	0				
4	254	0	0				

Ответ: 245937

2.11: САМОСТОЯТЕЛЬНО

Дана последовательность натуральных чисел.

Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не более чем на 6.**

Определите количество таких пар, для которых сумма чисел чётная. Исходные данные записаны в виде столбца электронной таблицы в файле **2_11.xls**.

2.11: САМОСТОЯТЕЛЬНО

Ответ: 2936

Подсказка: для первого числа предыдущих пар нет, поэтому ставим 0. Для **следующих 5** чисел формулу пишем с **зафиксированной** верхней границей: определяем количество пар среди чисел выше.

Для 7-го числа пишем формулу, которую будет тиражировать на весь столбец. В этой формуле верхняя граница **не зафиксирована**, она изменяется от ячейки к ячейке.

Формулы см. на следующем слайде.

2.11: САМОСТОЯТЕЛЬНО

1)

	A	B	C
1	123	1	0

2)

	A	B	C	D	E	F
1	123	1	0			
2	985	1	=COUNTIF(\$B\$1:B1;"="&B2)			
3	73	1	2			
4	254	0	0			
5	438	0	1			
6	856	0	2			
7	424	0				
8	925	1				
9	182	0				

3)

	A	B	C	D	E	F
1	123	1	0			
2	985	1	1			
3	73	1	2			
4	254	0	0			
5	438	0	1			
6	856	0	2			
7	424	0	=COUNTIF(B1:B6;"="&B7)			
8	925	1				
9	182	0				
10	421	1				

Эту формулу уже тиражируем на все ячейки ниже.

4)

	A	B	C	D	E	F
1	123	1	0			
2	985	1	1			
3	73	1	2			
4	254	0	0			

Formula bar: =SUM(C1:C1000)

	A	B	C	D	E	F
1	123	1	0		2936	

2.12: РАЗБОР


Дана последовательность натуральных чисел.

Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не более чем на 5.**

Определите **минимальную чётную** сумму среди таких пар. Исходные данные записаны в виде столбца электронной таблицы в файле **2_12.xls**.

2.12: РАЗБОР


В столбце В запишем остаток от деления числа из столбца А на 2:




	A	B	C	D	E	F
1	123	=MOD(A1;2)				
2	985	1				
3	73	1				
4	254	0				
5	438	0				
6	856	0				
7	424	0				
8	925	1				

2.12: РАЗБОР

В столбце С найдём минимальное число, с которым можно образовать пару. У первого числа пары нет, у следующих четырёх ищем среди всех чисел выше (верхняя граница зафиксирована).



	A	B	C	D
1	123	1	0	



	A	B	C	D	E	F	G
1	123	1	0				
2	985	1	=MINIFS(\$A\$1:A1;\$B\$1:B1;"&B2)				
3	73	1	123				
4	254	0	0				
5	438	0	254				

Здесь мы используем функцию MINIFS: она ищет минимальное число в списке чисел, которое удовлетворяет какому-то условию. Ищем среди чисел столбца А, при этом соответствующий остаток от деления в столбце В должен быть равен В2 (т.е. остатки от деления на 2 у чисел в столбце А совпадают).

2.12: РАЗБОР


Для остальных чисел верхняя граница НЕ зафиксирована (нет \$):

	A	B	C	D	E	F
1	123	1	0			
2	985	1	123			
3	73	1	123			
4	254	0	0			
5	438	0	254			
6	856	0	<code>=MINIFS(A1:A5;B1:B5;"="&B6)</code>			
7	424	0	254			
8	925	1	73			
9	182	0	254			

2.12: РАЗБОР

Пока мы нашли только минимальное подходящее число.
Теперь найдём сумму чисел в паре.

Если подходящее число равно 0 (т.е. ничто не подошло,
сделаем сумму очень большим числом, например, 100000).




	A	B	C	D	E	F
1	123	1	0	=IF(C1<>0;A1+C1;100000)		
2	985	1	123	1108		
3	73	1	123	196		
4	254	0	0	100000		


После тиражирования формулы в столбце D окажется минимальные суммы чисел в парах.

2.12: РАЗБОР

Осталось найти только самую маленькую сумму:



	A	B	C	D	E
1	123	1	0	100000	=MIN(D1:D1000)
2	985	1	123	1108	
3	73	1	123	196	
4	254	0	0	100000	



	A	B	C	D	E
1	123	1	0	100000	30
2	985	1	123	1108	
3	73	1	123	196	

Ответ: 30

2.13: САМОСТОЯТЕЛЬНО

Дана последовательность натуральных чисел. Рассматриваются всевозможные пары чисел, **порядковые номера которых отличаются не менее чем на 7**. Определите **максимальную сумму** среди таких пар, **кратную 5**. Исходные данные записаны в виде столбца электронной таблицы в файле **2_13.xls**.

Подсказка-1: во столбце В нужно хранить остаток от деления числа на 5, а в столбце С – остаток от деления **его пары** на 5. Обратите внимание: для остатка 1 пара – это остаток 4. **А для остатка 0 пара – это остаток 0.**

Подсказка-2: в столбце В будет использоваться функция MOD, в столбце С – функция IF.

2.13: САМОСТОЯТЕЛЬНО

Ответ: 2000

FLOOR.M... : X ✓ fx =MOD(A1;5)

	A	B
1	123	=MOD(A1;5)
2	985	0

FLOOR.M... : X ✓ fx =IF(B1=0; 0; 5-B1)

	A	B	C
1	123	3	=IF(B1=0; 0; 5-B1)
2	985	0	0

Продолжение на
следующем слайде

	A	B	C	D
1	123	3	2	0
2	985	0	0	0
3	73	3	2	0
4	254	4	1	0
5	438	3	2	0
6	856	1	4	0
7	424	4	1	0

2.13: САМОСТОЯТЕЛЬНО

FLOOR.M... : x ✓ f_x =MAXIFS(\$A\$1:A7;\$B\$1:B7;"="&C8)

	A	B	C	D	E	F
1	123	3	2	0	0	2000
2	985	0	0	0	0	
3	73	3	2	0	0	
4	254	4	1	0	0	
5	438	3	2	0	0	
6	856	1	4	0	0	
7	424	4	1	0	0	
8	925	0	0	0		
9	187	2	3	438	620	

Продолжение
на
следующем
слайде

	A	B	C	D	E	F
1	123	3	2	0	=IF(D1=0;0;D1+A1)	
2	985	0	0	0	0	
3	73	3	2	0	0	

2.13: САМОСТОЯТЕЛЬНО

D	E	F	G
0	0	=MAX(E1:E1000)	
0	0		
0	0		

	A	B	C	D	E	F
1	123	3	2	0	0	2000
2	985	0	0	0	0	
3	73	3	2	0	0	